/*
 *  Licensed to the Apache Software Foundation (ASF) under one
 *  or more contributor license agreements.  See the NOTICE file
 *  distributed with this work for additional information
 *  regarding copyright ownership.  The ASF licenses this file
 *  to you under the Apache License, Version 2.0 (the
 *  "License"); you may not use this file except in compliance
 *  with the License.  You may obtain a copy of the License at
 *
 *    https://www.apache.org/licenses/LICENSE-2.0
 *
 *  Unless required by applicable law or agreed to in writing,
 *  software distributed under the License is distributed on an
 *  "AS IS" BASIS, WITHOUT WARRANTIES OR CONDITIONS OF ANY
 *  KIND, either express or implied.  See the License for the
 *  specific language governing permissions and limitations
 *  under the License.
 */

plugins {
    id 'groovy'
    id 'java-library'
    id 'org.apache.grails.buildsrc.publish'
    id 'org.apache.grails.buildsrc.sbom'
}

version = projectVersion
group = 'org.apache.grails'

ext {
    pomTitle = 'Grails MongoDB JSON Views'
    pomDescription = 'Provides JSON Views for MongoDB to the Grails framework.'
    pomDevelopers = ['puneetbehl': 'Puneet Behl']
}

dependencies {

    implementation platform(project(':grails-bom'))

    api project(':grails-views-gson')

    implementation project(':grails-data-mongodb-core')
    implementation project(':grails-data-mongodb-bson')

    compileOnly 'org.slf4j:slf4j-nop' // Get rid of warning about missing slf4j implementation during compileGsonViews task
}

def templateSourceDir = layout.projectDirectory.dir('src/templates')
def compilationOutputDir = layout.buildDirectory.dir('classes/groovy/main')
sourceSets {
    main {
        groovy {
            // Add templates as source dir
            srcDirs = [templateSourceDir]
        }
    }
}

tasks.register('compileViews', JavaExec).configure {JavaExec it ->
    it.inputs.dir(templateSourceDir)
    it.outputs.dir(compilationOutputDir)
    it.mainClass = 'grails.plugin.json.view.JsonViewCompiler'
    it.classpath = configurations.compileClasspath
    it.args(templateSourceDir.asFile, compilationOutputDir.get().asFile, '17', ' ', ' ', 'none', 'UTF-8')
}

// This is needed to trigger compilation of the views
tasks.named('classes').configure {
    it.dependsOn 'compileViews'
}

apply {
    from rootProject.layout.projectDirectory.file('gradle/java-config.gradle')
    from rootProject.layout.projectDirectory.file('gradle/code-style-config.gradle')
    from rootProject.layout.projectDirectory.file('gradle/mongodb-test-config.gradle')
    from rootProject.layout.projectDirectory.file('gradle/docs-config.gradle')
}
